

# VIDEO BASIC

20 VIDEOLEZIONI DI BASIC  
PER IMPARARE COL C16



**GRUPPO  
EDITORIALE  
JACKSON**

*Cos'è un robot*

*Come insegnare a un robot*

*Il futuro della robotica*

*Progetto e codifica  
di un carattere*

*READ DATA RESTORE*

*DO LOOP EXIT*

*WHILE UNTIL*

*Calendario perpetuo*

*Assegnazione dati*

*Videogioco n° 11*

11

C16

commodore 16 plus 4



## VIDEOBASIC C 16

Pubblicazione quattordicinale  
edita dal Gruppo Editoriale Jackson

### Direttore Responsabile:

Giampietro Zanga

### Direttore e Coordinatore

Editoriale: Roberto Pancaldi

Autore: Softidea -

Via Indipendenza 88-90 - Como

### Redazione software:

Giuliano Cremonesi

Francesco Franceschini

### Progetto grafico:

Studio Nuovidea - via Longhi, 16 - Milano

### Impaginazione:

Moreno Confalone

### Illustrazioni:

Cinzia Ferrari, Silvano Scolari

### Fotografie:

Marcello Longhini

### Distribuzione: SODIP

Via Zuretti, 12 - Milano

### Fotocomposizione: Lineacomp S.r.l.

Via Rosellini, 12 - Milano

### Stampa: Grafika '78

Via Trieste, 20 - Pioltello (MI)

### Direzione e Redazione:

Via Rosellini, 12 - 20124 Milano

Tel. 02/6880951/5

Tutti i diritti di riproduzione e pubblicazione di  
disegni, fotografie, testi sono riservati.

© Gruppo Editoriale Jackson 1986.

Autorizzazione alla pubblicazione Tribunale di  
Milano n° 422 del 22-9-1984

Spedizione in abbonamento postale Gruppo II/70  
(autorizzazione della Direzione Provinciale delle  
PPTT di Milano).

Prezzo del fascicolo L. 8.000

Abbonamento comprensivo di 5 raccoglitori L. 165.000

I versamenti vanno indirizzati a: Gruppo

Editoriale Jackson S.p.A. - Via Rosellini, 12

20124 Milano, mediante emissione di assegno

bancario o cartolina vaglia oppure

utilizzando il c.c.p. n° 11666203.

I numeri arretrati possono essere

richiesti direttamente all'editore

inviando L. 10.000 cdu. mediante assegno

bancario o vaglia postale o francobolli.

Non vengono effettuate spedizioni contrassegno.

---

## SOMMARIO

---

### HARDWARE ..... 2

Cos'è un robot. Come insegnare  
a un robot. Sviluppi futuri.

### IL LINGUAGGIO ..... 10

Definizione di caratteri.  
READ, DATA, RESTORE, DO, LOOP,  
WHILE, UNTIL, EXIT.

### LA PROGRAMMAZIONE ..... 22

Calendario. Definizione  
di un carattere. Indicativo  
presente verbi in ARE.

### VIDEOESERCIZI ..... 32

---

## Introduzione

*In questa lezione vedremo innanzitutto  
come funzionano, come apprendono  
e come lavorano i robot, accennando  
anche alle affascinanti prospettive  
e possibilità che questi dispositivi  
promettono di aprire nella vita di tutti  
i giorni.*

*Scopriremo quindi in quale modo sia  
possibile inserire permanentemente  
in un programma informazioni utili  
o ricorrenti, utilizzando le istruzioni  
READ, DATA e RESTORE.  
Per finire, vedremo le istruzioni dei  
cicli strutturati (DO, LOOP, WHILE,  
UNTIL, EXIT) un esempio sulla tecnica  
da adottare per personalizzare le  
visualizzazioni in uscita, imparando  
come definire e memorizzare nuovi  
caratteri all'interno del nostro  
computer.*



**GRUPPO EDITORIALE  
JACKSON**

DIVISIONE GRANDI OPERE


# HARDWARE

## Cos'è un robot

Sin da prima dell'avvento dei calcolatori elettronici, una delle maggiori aspirazioni dell'uomo fu quella di riuscire ad inventare e costruire nuove macchine, che potessero sostituirlo nei lavori più faticosi, pericolosi o ripetitivi.

La storia ce lo insegna: ciascun gradino della scala del progresso è stato (ed è tuttora) normalmente caratterizzato da una grande e fondamentale invenzione o scoperta (come il fuoco, il ferro, il vapore o l'elettricità), che - inizialmente realizzata per risolvere determinati problemi - una volta applicata e sviluppata ci ha procurato (e ci procura) nuovi e magari inaspettati progressi e perfezionamenti. Il calcolatore elettronico, per quanto non possa essere considerato una invenzione nel senso stretto della parola, non sfugge a questa regola. Nato per "calcolare", il computer è diventato la base per una infinità di nuove applicazioni, che su di esso si basano e dal quale dipendono strettamente. Tra queste la robotica, cioè la scienza che si occupa dell'automatizzazione del lavoro attraverso macchine controllate da computer. Robot è un termine che nella mente di molte persone suscita tuttora parecchi dubbi e perplessità: il ricordo di uomini metallici e di macchine parlanti, tanto cari alla fantascienza più

commerciale, ne rendono forse difficile l'esatta comprensione. Per prima cosa cerchiamo allora di spiegarne il significato. Un robot non è altro che una macchina automatica, la quale sotto il controllo di un computer, svolge un lavoro prestabilito: ad esempio verniciare un'automobile, avvitare dei bulloni o saldare delle lamiere. Il grosso vantaggio di un robot rispetto ad una normale macchina automatica è che, essendo collegato ad un computer, può essere programmato - con modifiche di solito non





# HARDWARE

molto dispendiose - per svolgere lavori diversi (non troppo diversi, però!).

Alla programmabilità si aggiunge un altro pregio, grazie ai progressi della microelettronica i robot vengono infatti dotati di particolari dispositivi, chiamati sensori, che - per quanto ancora ben

lontani dai sensi dell'uomo - permettono ad un robot di riconoscere il verificarsi di certe situazioni, particolari od anomale. Facciamo un esempio. Supponiamo di dover avvitare con un bullone due pezzi di metallo appositamente predisposti. Se utilizziamo per questa operazione un robot, possiamo metterci al riparo da ipotetiche evenienze, dovute al cattivo allineamento dei due pezzi, al difettoso fissaggio del bullone od all'errata dimensione di uno qualsiasi dei componenti. Fornendo alla macchina degli appositi sensori e programmando opportunamente il calcolatore di controllo è possibile arrestare la macchina al verificarsi di uno di questi casi. Una macchina normale, in presenza di uno di questi casi anomali,

proseguirebbe imperterrita nel proprio lavoro, provocando magari la messa in produzione di un pezzo difettoso o, peggio ancora, la rottura della macchina stessa. L'introduzione del robot inserisce quindi nella catena di montaggio un elemento di controllo e di verifica, oltre che di produzione. Molti robot possono inoltre svolgere lavori che per le persone sarebbero pericolosi o sgradevoli, come misurare la radioattività in una centrale nucleare, disinnescare una bomba o operare in atmosfere inquinate.




# HARDWARE

## Come insegnare ad un robot

La maggior parte dei robot esistenti lavora ed opera in fabbriche in cui tutto è organizzato intorno a loro: di solito debbono prelevare dei pezzi da nastri trasportatori, lavorarli seguendo una certa successione di

operazioni elementari e depositarli infine su un altro nastro trasportatore, che provvede ad avviarli verso altre lavorazioni. Le prestazioni di un robot dipendono



# HARDWARE

comunque da molti fattori, non ultimo quello economico; esistono quindi (e sono la maggioranza, visto il loro minor costo e la relativa semplicità) robot,

costruiti per un uso abbastanza generico, che di volta in volta vengono adattati per svolgere altre mansioni più specifiche. I robot più comuni sono quelli cosiddetti "a braccio": la loro meccanica è infatti costruita prendendo a modello le braccia dell'uomo. Tale disposizione permette - proprio come per le braccia vere - una notevolissima libertà e capacità di movimenti, consentendo alle "mani" (che normalmente sono pinze comandate da motori elettrici) di operare in posizioni altrimenti inaccessibili. Attualmente si sta anche studiando la maniera di dotare queste "mani" di una sorta di senso del tatto, consentendo così una presa adeguata al peso ed alla solidità dell'oggetto da sollevare (finora nessun robot riesce a distinguere un pezzo di ferro a forma d'uovo da un uovo vero, con risultati facilmente intuibili). Progettare, costruire ed insegnare ad un robot è pertanto un'operazione per niente semplice: occorre innanzitutto analizzare e suddividere l'azione che il robot

dovrà svolgere in tutti i possibili ed immaginabili passaggi elementari (ti ricordi gli algoritmi? ...), classificandone così le diverse necessità di movimento. In base a questi movimenti - ed in funzione dello sforzo necessario per compiere ciascuna azione - si deciderà il numero e la posizione dei bracci meccanici, dei giunti (cioè degli snodi tra i singoli bracci) e dei relativi motori.

Per eseguire il proprio lavoro il robot dovrà inoltre essere provvisto di sensori di vario tipo (fotoelettrici, elettronici o meccanici), appropriati al genere di lavorazione richiesto. Attraverso una o più interfacce questi sensori dovranno quindi inviare le informazioni codificate all'elaboratore di controllo, addetto alla direzione ed al coordinamento dei vari movimenti, che provvederà così a comandare l'esecuzione, la modifica o l'arresto delle operazioni previste in origine.

Il computer di controllo andrà pertanto programmato in modo adeguato a queste necessità, tenendo in debito conto tutte le possibilità di imprevisti

# HARDWARE

od anomalie. E questo per decine o centinaia di singole azioni.

Ancora una volta ti puoi rendere conto di quanto sia importante - prima di passare alla fase vera e propria di programmazione - valutare nei suoi più piccoli particolari tutto il complesso di operazioni, indipendentemente dal fatto che siano importanti o insignificanti che il microprocessore del robot - una volta divenuto operativo - sarà sempre chiamato a controllare.

In un robot, soprattutto se di tipo industriale, questa fase di analisi è di particolare difficoltà, visto che di solito un unico microprocessore non è in grado di affrontare da solo l'intera mole di lavoro che il robot deve poter svolgere. Molto spesso (per non dire sempre) si

rende quindi necessario programmare una serie di microprocessori che sviluppino tutte le operazioni particolari (come, per esempio, muovere un braccio o afferrare un oggetto), coordinandoli attraverso un'ulteriore, grossa unità centrale di controllo, che riesca a seguire ogni momento della fase di lavorazione.

Come ben puoi immaginare, se già è difficile programmare un solo calcolatore, programmarne 10 o 15 (perché possano oltre tutto lavorare in collaborazione tra loro) diventa un'impresa estremamente delicata.

## Sviluppi futuri

La robotica è una scienza molto giovane, si può dire appena agli inizi; nel suo futuro vi potranno quindi essere molti sviluppi e miglioramenti, grazie anche ai numerosi progetti di ricerca attualmente in corso in tutto il mondo.

In questo momento una delle strade più battute (a onor del vero, non solo nella robotica) è quella della cosiddetta


“intelligenza artificiale”. Detta in breve, l'intelligenza artificiale sta cercando di dotare i calcolatori di una sorta di intelligenza autonoma ed automatica, capace di apprendere e ricordare gli eventi passati per




# HARDWARE



# HARDWARE



# HARDWARE



applicarli in quelli futuri (proprio come succede per i bambini che una volta scottati non cercano più di toccare il fuoco). Finora i risultati non sono stati molto incoraggianti, ma sembra che entro pochi anni si potrà cominciare a vedere qualcosa. Comunque, al di là di qualsiasi timore che può incutere l'idea di macchine intelligenti, un robot capace di apprendere dall'ambiente circostante consentirebbe di risolvere molti dei problemi legati alla progettazione ed alla programmazione del robot stesso, evitando quella lunga e complessa fase di studio ed analisi che abbiamo visto essere uno degli scogli più impegnativi nella costruzione e nell'installazione di tutti i dispositivi automatici.



Anche se gli sviluppi della robotica sembrano promettere per il prossimo futuro cose molto interessanti, al momento attuale, permangono ancora diversi problemi: i robot non sanno muoversi come si vorrebbe (nessuno a tutt'oggi è riuscito a far camminare un robot come un uomo), costano abbastanza cari (soprattutto i robot industriali) e non riescono ad essere adattati per un uso universale. In commercio ne esistono comunque già di tantissimi tipi, addirittura anche per uso personale. Lo scopo di questi ultimi, più che altro, è però rivolto all'insegnamento dei rudimenti della robotica e solo raramente consentono di ottenere prestazioni adatte per un'applicazione reale. Essi permettono comunque di apprendere con facilità, e con una spesa relativamente limitata, il funzionamento e la programmazione dei robot, essendo molto spesso collegabili ai più diffusi personal computer per il controllo dei movimenti.

# LINGUAGGIO

## Definizione caratteri

Accade talvolta che, per necessità o per divertimento, si desideri modificare uno o più caratteri tra quelli normalmente disponibili sul tuo C16. Per fare ciò, è necessario eseguire una

semplice operazione, meglio nota come definizione dei caratteri, che consente di creare un nuovo insieme di simboli da far corrispondere a ciascuno dei tasti presenti sulla tastiera.



Un carattere è racchiuso in una matrice di 8 punti per 8. Quando vuoi rappresentarlo nei dettagli è necessario operare su una matrice di punti maggiori.

Nell'esempio puoi notare come per riprodurre dettagliatamente un alfabeto particolare si sia fatto ricorso ad una matrice di 16 x 16 (4 caratteri).

# LINGUAGGIO

Così, se per esempio volessimo poter disporre sulla tastiera delle lettere dell'alfabeto greco, basterebbe che "insegnassimo" al computer la forma di ciascuno dei nuovi caratteri, in modo che al tasto "A" corrisponda la "alfa", al tasto "B" la "beta", e così via.

Tu sai già cosa fa il C16 per essere in grado di visualizzare tutti i caratteri che puoi di solito vedere sul video, e sai anche che le forme dei vari caratteri (costituite da serie di 8 byte per ogni carattere) si trovano gelosamente custodite nella memoria a sola lettura (ROM), perché non vengano perse ogni volta che spegni il calcolatore. Quando premi un tasto, questo corrisponde quindi ad un indirizzo della ROM, la quale contiene la sequenza di informazioni che il computer desidera conoscere perché il carattere possa apparire sul video. Dal momento che la memoria ROM non può essere alterata in alcun modo, la prima cosa da fare, se si vogliono

definire dei nuovi caratteri, è perciò quella di trasferire tutte le immagini dei vecchi caratteri in una zona accessibile anche alla scrittura (cioè nella memoria RAM). A questo punto tutto è predisposto per • accettare le eventuali modifiche che desidereremo effettuare ai vari caratteri. Per imparare a definire un nuovo insieme di caratteri è però necessario esaminare innanzi tutto come si deve fare per creare un singolo carattere. Ciascun carattere è composto da una combinazione di 64 punti (ottenuta utilizzando gli 8 bit di 8 byte), disposti su 8 righe ed 8 colonne. Ecco un esempio, rappresentato da una sequenza di questo tipo:

BINARIO	DEC.	
0 0 1 1 1 1 0 0	60	
0 1 1 0 0 1 1 0	102	
0 1 1 0 1 1 1 0	110	
0 1 1 0 1 1 1 0	110	
0 1 1 0 0 0 0 0	96	
0 1 1 0 0 1 1 0	102	
0 0 1 1 1 1 0 0	60	
0 0 0 0 0 0 0 0	0	

# LINGUAGGIO

dove ogni 1 o 0 rappresenta rispettivamente un punto (pixel) acceso o spento. Per un computer, acceso o spento significa però valore 1 o 0: basterà allora introdurre nelle locazioni adibite alla

definizione di quel carattere gli otto numeri binari, letti per riga, che definiscono le varie combinazioni di pixel. Nel nostro esempio, prendendo la prima riga (00111100) otteniamo un certo numero binario, che, convertito in decimale (60), ci fornisce uno degli otto valori da inserire nelle locazioni della memoria che specificheranno la descrizione e la struttura di quel carattere. Per chiarezza questo valore è stato scritto sulla destra della medesima riga. La stessa operazione sulle altre sette righe ci procura infine i restanti valori, necessari per definire completamente tutto il carattere. Gli otto numeri che derivano da  sono pertanto: 60, 102, 110, 110, 96, 102, 60 e 0. A questo punto li si potrà inserire nella memoria. Quando il computer dovrà visualizzare questo carattere preleverà allora gli otto numeri dalla memoria e li visualizzerà sullo schermo. Questo è tutto. Il primo passo da eseguire nella definizione di un carattere è perciò quello di disegnare una serie di

64 quadretti nei quali definire i pixel da accendere o da spegnere. Creando un qualsiasi carattere è buona norma cercare di non usare mai uno 0 od un 1 isolato: infatti, adoperando un televisore di qualità non ottima, è probabile che il punto corrispondente non venga visualizzato. Fatta la prima volta, la via da seguire per impostare altri caratteri è sempre la stessa: cambieranno soltanto i numeri da inserire nel computer (ogni carattere è chiaramente definito da una particolare ed unica sequenza di punti luminosi) e gli indirizzi delle locazioni nelle quali introdurre questi numeri. Alla fine l'insieme dei caratteri sarà stato definito e potrà essere utilizzato a piacimento da qualsiasi programma. Per recuperare il vecchio insieme di caratteri sarà sufficiente spegnere il computer, riaccendendolo dopo qualche secondo: questa operazione cancellerà tutte le modifiche effettuate nella memoria RAM, ripristinando gli indirizzi della ROM che corrispondono ai soliti tasti.

## READ/DATA


Le istruzioni READ e DATA permettono di leggere dei dati all'interno di un programma, evitando così di doverli impostare manualmente dalla tastiera.

L'introduzione dei dati da elaborare avviene inoltre senza il momentaneo arresto del

programma, al contrario di quanto accade per l'istruzione INPUT.

Con ogni probabilità ti starai però chiedendo quando mai potrà servirti una simile struttura, visto che finora sei riuscito benissimo a farne a meno.

Un esempio ti chiarirà subito le idee.



# LINGUAGGIO

Accade molto spesso che un programma richieda come azione preliminare dell'esecuzione la cosiddetta inizializzazione delle variabili, cioè l'inserimento di alcuni

specifici valori in determinate variabili (per esempio i nomi e le durate dei singoli mesi dell'anno).

Per fare questa operazione si hanno a disposizione tre possibili alternative:


— utilizzare una lunga serie di LET all'inizio del programma;

— richiedere ad ogni esecuzione, utilizzando delle INPUT, i valori da assegnare alle singole

variabili;

— adoperare READ e DATA.

Scartiamo subito la prima soluzione: richiederebbe troppo lavoro durante la battitura del programma e farebbe occupare troppa memoria (ricorda che ogni istruzione conservata nel computer occupa una certa porzione di memoria). La seconda soluzione è già più accettabile. Nel



# LINGUAGGIO

caso dei mesi si potrebbe scrivere:

```
10 DIM M$ (12), G (12)
20 FOR I = 1 TO 12
30 INPUT M$ (I), G (I)
40 NEXT I
```

e con queste poche istruzioni ce la saremmo cavata.

Il problema è però risolto solo parzialmente: ad ogni RUN dovremmo infatti metterci di buona lena a battere GENNAIO, 31, FEBBRAIO, 28, ... ecc., impostando cioè dei valori che, tutto sommato, tra un'esecuzione e l'altra non subiscono alcuna modifica e che quindi

sarebbe comodo conservare in permanenza nel programma. Ciò significa che ogni volta che utilizziamo questo programma dobbiamo assegnare da tastiera tutte le 24 informazioni che ci vengono richieste. Lavoro lungo, noioso ed inutile.

Adottando READ e DATA avremo invece:

```
10 DIM M$ (12), G (12)
20 FOR I = 1 TO 12
30 READ M$ (I), G (I)
40 NEXT I
50 DATA GENNAIO, 31, FEBBRAIO, 28, MARZO, 31, APRILE, 30
60 DATA MAGGIO, 31, GIUGNO, 30, LUGLIO, 31, AGOSTO, 31
70 DATA SETTEMBRE, 30, OTTOBRE, 31, NOVEMBRE, 30,
DICEMBRE, 31
```



In pratica queste istruzioni ordinano al calcolatore di fare le stesse cose delle linee viste prima, con la sola differenza che le singole variabili degli array M\$ ( ) e G ( ) devono essere lette (READ) non più dalla tastiera - come accadeva prima a causa dell'INPUT - ma dalle righe DATA.

I dati vengono quindi inseriti nel calcolatore una sola volta e, cosa ancora più importante, chi usa il programma

# LINGUAGGIO

non è tenuto a conoscere ed a battere questi dati (prova a pensare ai nomi di tutte le squadre del campionato di calcio o ad altre serie di dati, magari meno noti dei nomi dei mesi dell'anno). In questo modo è possibile mantenere - conservati all'interno del programma - dei valori che altrimenti andrebbero tutte le volte dispersi con lo spegnimento del calcolatore o con la riesecuzione del programma stesso. Le istruzioni DATA contengono i diversi valori che si vogliono assegnare. Esse permettono di riunire tutti i dati in un unico punto del programma, di solito all'inizio o alla fine, dove sono più leggibili. L'istruzione READ consente di leggere i dati che sono memorizzati nelle DATA. È ovvio che si debba fare estrema attenzione affinché il tipo della variabile dell'istruzione READ corrisponda al

contenuto dell'istruzione DATA, proprio come succede negli INPUT da tastiera: non si dovrà mai verificare che una variabile numerica possa contenere un valore alfanumerico o che in una variabile di tipo stringa si trovi un valore numerico. Questo tipo di errore è sempre in agguato!

Le linee DATA del listato precedente sono tre solo per motivi di leggibilità: avremmo infatti potuto metterne un numero diverso anche non necessariamente consecutive. Infatti se vi sono più linee DATA, in punti diversi del programma, queste sono lette di seguito, come se si trattasse di un'unica linea.

## Esempi

```
10 READ A, B
20 DATA 3, 5
```

Alle variabili A e B vengono assegnati i valori 3 e 5.

```
10 READ A, B, C
20 DATA 3, 5
```

Non ci siamo: alla variabile C non corrisponde alcun valore e si avrà il messaggio di errore OUT OF DATA.

```
5 READ A, A$, C
10 DATA 3, 4, 5
```

A, A\$ e C assumono rispettivamente valore 3, "4" e 5.

Poiché la seconda variabile (A\$) è di tipo stringa, 4 viene considerato valore stringa quindi inserito in memoria come carattere e non come numero.

```
5 READ C$, A, D$
10 DATA PIPPO, PLUTO
```

In questo caso gli errori sono due: la variabile A non può assumere

# LINGUAGGIO

valore PLUTO (essendo di tipo numerico) e la variabile D\$ è sprovvista del corrispondente DATA.

```
5 FOR I = 1 TO 6
8 READ K$
10 PRINT K$
13 NEXT I
20 DATA FRANCO, MARIO
30 DATA MATTEO, CARLO
40 DATA PIPPO, GIACOMO
```

La variabile K\$ assumerà via via i 5 valori specificati nelle istruzioni DATA. Il comando PRINT K\$, inserito nel ciclo FOR, provocherà pertanto la stampa di FRANCO, MARIO, ..., GIACOMO. Nota come la variabile K\$ sia dello stesso tipo (stringa) dei valori che dovrà assumere in seguito alle READ.

```
5 SCNCLR
10 PRINT "SISTEMA SOLARE" : PRINT
15 FOR K = 1 TO 9
20 PRINT
25 READ P$, D
30 PRINT "IL PIANETA "; P$
35 PRINT "HA DIAMETRO DI"; D; "KM"
40 NEXT K
45 DATA MERCURIO, 4880, VENERE, 12096
50 DATA TERRA, 12740, MARTE, 6780
55 DATA GIOVE, 141560, SATURNO, 120800
60 DATA URANO, 51000, NETTUNO, 49300
65 DATA PLUTONE, 7000
```

Questo, più che un esempio, è una dimostrazione dell'utilità di READ ... DATA. Con poche istruzioni è infatti possibile inserire in permanenza delle informazioni che non cambieranno mai tra un'esecuzione e l'altra, consentendo un considerevole risparmio sia di spazio che di tempo.

## Sintassi dell'istruzione DATA

---

DATA costante [ , costante ] [ , ... ]

---

## Sintassi dell'istruzione READ

---

READ variabile [ , variabile ] [ , ... ]

---

## RESTORE

Per controllare l'ordine secondo cui vengono prelevati i valori delle DATA possiamo immaginare che all'interno della memoria esista una specie di indice (chiamato anche puntatore) che segnali - di volta in volta - qual è il successivo valore che può essere letto con un'istruzione READ. Ogni volta che il computer legge un


qualsiasi elemento dalle linee DATA questo puntatore viene perciò fatto spostare sull'elemento successivo, in modo che l'interprete BASIC sappia sempre sin dove è arrivata la lettura dei vari termini delle DATA. Ad ogni READ corrisponderà quindi un automatico avanzamento del puntatore delle DATA.

Talvolta può comunque essere molto utile avere la possibilità, nel corso del programma, di

accedere alle stesse informazioni più di una volta.

L'istruzione RESTORE serve appunto per riportare il puntatore interno alla prima linea DATA del programma rendendo nuovamente disponibili le costanti come se non fossero mai state lette.

Nel caso invece, il valore da rileggere non fosse il primo, è necessario indicare, dopo RESTORE, il numero della linea da cui iniziare a prelevare gli elementi.



# LINGUAGGIO

```
10 DATA DO, RE, MI, FA, SOL
20 DATA GIALLO, VERDE
30 FOR I = 1 TO 7 : READ A$
40 PRINT A$ : NEXT I
```

Il programma fino a questo punto stampa tutti gli elementi delle linee DATA.

Cioè le note e i colori.

Il puntatore è posizionato dopo la costante "verde".

Ecco come proseguire per ristampare solamente i colori:

```
50 RESTORE 20: REM POSIZIONE DEL
  PUNTATORE SUL "GIALLO" DELLA LINEA 20
60 FOR I = 1 TO 2 : READ A$
70 PRINT A$ : NEXT I
```

Le linee 60 e 70 provvedono alla ristampa dei colori "GIALLO" e "VERDE"

Quando viene incontrata, l'istruzione RESTORE - indipendentemente dal numero di elementi che

sono già stati letti - fa quindi in modo che la successiva istruzione READ torni indietro a leggere il primo elemento nella lista dei DATA, proprio come accade la prima volta. Così, il seguente programma

```
10 FOR I = 1 TO 1000
20 READ A
25 PRINT A
30 RESTORE
40 NEXT I
50 DATA 27, 10, 60
```

avrà come unico risultato la stampa per 1000 volte del valore 27, visto che il RESTORE posto alla linea 20 impedisce al puntatore delle DATA di avanzare al termine successivo. Prova adesso a togliere la riga 30 ed a eseguire il programma: se avevi dei dubbi sul funzionamento di RESTORE, li chiarirai subito.

## Sintassi dell'istruzione

---

RESTORE [NUMERO LINEA]

---

# LINGUAGGIO

## **DO, LOOP, WHILE, UNTIL, EXIT**

L'insieme di queste istruzioni serve a definire e controllare un ciclo di programma in modo analogo a quello svolto dal FOR TO STEP NEXT. Ha però un grosso vantaggio rispetto a quest'ultimo: non richiede infatti di conoscere a priori i valori limite del contatore.

La traduzione in italiano delle parole chiave potrebbe essere la seguente: ESEGUI (DO), RIPETI (LOOP), FINTANTO CHE (WHILE), FINO A QUANDO (UNTIL), ESCI (EXIT).

La struttura del ciclo fa eseguire le istruzioni poste tra DO e LOOP. Se le operazioni logiche (Booleane) susseguenti WHILE o UNTIL non sono verificate, l'esecuzione delle istruzioni tra DO e LOOP continuerà all'infinito. Quando all'interno del ciclo viene incontrata l'istruzione EXIT, il controllo del programma verrà affidato alla prima istruzione posta dopo LOOP.

I cicli DO possono essere nidificati seguendo la regola dei FOR-NEXT.

Nel caso venga specificato il parametro UNTIL o WHILE il programma continuerà ad eseguire il ciclo fino a quando la condizione non sarà verificata.

La distinzione tra WHILE e UNTIL è la seguente: con WHILE il ciclo viene ripetuto FINTANTO CHE la condizione RESTA come quella specificata; con UNTIL, invece, FINO a QUANDO la

# LINGUAGGIO

condizione DIVENTA  
come quella specificata.  
Un esempio di  
condizione logica  
potrebbe essere  $A=1$  o  
 $G>23$ .

## Esempi

```
10 X=25
20 DO UNTIL X=0
30 X=X-5
40 PRINT "X="; X
50 LOOP
60 PRINT "Fine del ciclo"
```

Le istruzioni della linea  
30 e 40 sono eseguite  
fino a quando X diventa  
uguale a 0 e il  
programma riprende  
dalla linea 60 posta  
dopo l'istruzione LOOP.

```
10 DO WHILE a$ = " "
20 GET a$ : LOOP
30 PRINT "Il tasto è stato premuto"
```

Grazie a WHILE si  
attende fintanto che,  
grazie ad un evento  
esterno (la pressione di  
un tasto) non si verifica  
più a condizione logica  
impostata ( $a\$ = " "$ ).  
L'istruzione GETKEY a\$  
si comporta esattamente  
come la linea 10  
dell'esempio.

## Sintassi dell'istruzione


```
DO [UNTIL condizione/WHILE condizione]
istruzioni [EXIT] LOOP
```

# PROGRAMMAZIONE

## Calendario

Il primo esempio della nostra lezione è un'utile applicazione sull'uso delle istruzioni READ/DATA. Come risultato dell'esecuzione di questo programma si ha infatti la stampa del

calendario di un qualsiasi anno a partire dal 1985. Vediamo in che modo. Per prima cosa occorre conoscere quale giorno della settimana corrisponde al primo



# PROGRAMMAZIONE

giorno dell'anno prescelto. Ciò può essere fatto in modo molto semplice, tenendo cioè presente una data particolare: per esempio, il 1 gennaio 1984, che

era una Domenica. A questo punto è sufficiente calcolare il numero di giorni che intercorrono tra l'inizio dell'anno prescelto e l'inizio del 1984 (tenendo naturalmente conto degli anni bisestili) per conoscere automaticamente in quale giorno della settimana cade il 1 gennaio dell'anno prescelto.

Prendiamo, per esempio, il 1986. Tra il 1 gennaio 1984 ed il 1 gennaio 1986 vi sono 731 giorni (365 + 366 perché l'84 era bisestile).

Adesso basta quindi dividere 731 per 7

$$731 : 7 = 104 \text{ con avanzo di } 3$$


per sapere che il 1 gennaio '86 è un mercoledì. Il 3 di avanzo ci dice infatti che l'inizio dell'86 è 3 giorni dopo l'inizio dell'84: quindi (con scrittura forse non molto corretta)

$$\text{DOMENICA} + 3 = \text{MERCOLEDI}$$

A questo punto la strada è tutta in discesa: basterà stampare uno dopo l'altro i vari mesi dell'anno, curando che l'impaginazione sia corretta, per ottenere il nostro calendario.


# PROGRAMMAZIONE

Qui di seguito puoi vedere lo schema a blocchi generale del programma ed il corrispondente listato BASIC.



# PROGRAMMAZIONE

```
10 SCNCLR
20 DIMM(12),M$(12),G$(7)
30 REM LEGGI I NOMI DEI GIORNI
40 FORI=1TO7 : READG$(I) : G$(I)=LEFT$(G$(I),1) : NEXTI
50 REM LEGGI I NOMI E LA DURATA DEI MESI
60 FORI=1TO12: READM(I), M$(I) : NEXTI
70 REM
80 REM CALCOLA IL PRIMO GIORNO DELL'ANNO
90 INPUT "ANNO:"; AN
100 LETNG=0
```



# PROGRAMMAZIONE

```
110 IFAN<1985THENPRINT"❖❖";:GOTO90
120 SCNCLR
130 REMTROVA QUANTI GIORNI SONO PASSATI TRA IL 1 GEN. 1984 ED IL
    GEN. DELL'ANNO DESIDERATO
140 FORI=1984+1TOAN
150 LETNG=NG+365
160 REM SE L'ANNO È BIESTILE AGGIUNGI UN GIORNO
170 IF(I/4=INT(I/4))AND(I<>AN)THENLETNG=NG+1
180 NEXTI
190 REM ORA TROVA QUALE GIORNO DELLA SETTIMANA
    È IL PRIMO DELL'ANNO
200 LETNG=NG-(INT(NG/7))*7
210 LETJ=NG+1
220 IFAN/4=INT(AN/4)THENLETM(2)=29
230 FORI=1TO12
240 PRINTSPC(4);M$(I);" ";AN
250 PRINT
260 REM STAMPA LE INIZIALI DEI GIORNI
270 FORK=1TO7
280 PRINT " ";G$(K);" ";
290 NEXTK
300 PRINT : PRINT
310 IFJ=1THEN340
320 PRINTSPC((J-1)*3)
330 REM STAMPA IL CALENDARIO DEL MESE
340 FORK=1TOM(I)
350 A$=STR$(K):IFLEN(A$)=2THENA$=" "+A$
360 IFLEN(A$)=3THENA$=RIGHT$(A$,2)
370 PRINTA$;" ";
380 LETJ=J+1
390 IFJ=8THENLETJ=1:PRINT
400 NEXTK
410 PRINT : PRINT : PRINT
420 NEXTI
430 END
440 DATALUNEDI, MARTEDI, MERCOLEDI, GIOVEDI, VENERDI,
    SABATO, DOMENICA
450 DATA31, GENNAIO, 28, FEBBRAIO, 31, MARZO, 30, APRILE,
    31, MAGGIO, 30, GIUGNO
460 DATA31, LUGLIO, 31, AGOSTO, 30, SETTEMBRE, 31, OTTOBRE, 30,
    NOVEMBRE, 31, DICEMBRE
```

# PROGRAMMAZIONE

## Definizione di un carattere

Tu sai già come il C16 possa visualizzare tutti i caratteri che puoi vedere sul video, e sai anche che le immagini di questi caratteri (una serie di 8 byte per ogni carattere) sono gelosamente custodite in ROM (memoria a sola lettura), perchè non vengano perse ogni volta che spegni il calcolatore.

Può sembrare, quindi che ogni volta che verrà premuto il tasto "@" del C 16 il video mostrerà una @. Invece il C16 ti dà anche la possibilità di creare uno per uno tutti i caratteri. Proviamo a definire un carattere che assomigli a un omino! Usando la stessa tecnica, con cui sono disegnati normalmente i



# PROGRAMMAZIONE

caratteri, costruiamo una matrice 8 per 8 e riempiamola in modo da ottenere il simbolo voluto, usando 0 per i punti spenti e 1 per quelli accesi.

POTENZE DI 2								ADDIZIONE	VALORE DECIMALE DA "POKARE"
128	64	32	16	8	4	2	1		
0	0	0	1	1	0	0	0	→ 16 + 8	= 24
0	0	0	1	1	0	0	0	→ 16 + 8	= 24
0	0	1	1	1	1	0	0	→ 32 + 16 + 8 + 4	= 60
0	1	0	1	1	0	1	0	→ 64 + 16 + 8 + 2	= 90
1	0	0	1	1	0	0	1	→ 128 + 16 + 8 + 1	= 153
0	0	1	0	0	1	0	0	→ 32 + 4	= 36
0	0	1	0	0	1	0	0	→ 32 + 4	= 36
0	1	1	0	0	1	1	0	→ 64 + 32 + 4 + 2	= 102

Calcolo dei valori necessari per definire il nuovo carattere.

Ora ci tocca fare un pò di conti: la prima riga (cioè il primo byte del carattere) contiene il numero binario 00011000, cioè 24 decimale. Facendo lo stesso calcolo per le altre righe otteniamo una serie di 8 numeri cioè 24, 24, 60, 90, 153, 36, 36 e 102.

Poichè come sai non è possibile modificare un carattere nella ROM è necessario costruire i caratteri da sostituire in RAM e successivamente "ingannare" il computer sostituendogli il numero (puntatore) che gli dice dove andare a leggere il set dei caratteri (linee 60-70).

Obiettivo del programma è sostituire il carattere @ con quello dell'omino  progettato in precedenza. Aiutandoti con i commenti (linee REM) cerca di seguire passo dopo passo la tecnica utilizzata per creare un carattere.

# PROGRAMMAZIONE

```
10 REM AGGIORNA I PUNTATORI A FINE MEMORIA
20 POKE56,56 : POKE55,0 : CLR : PRINTCHR$(142)
30 REM
40 REM MODIFICA PUNTATORI PER PRELIEVO CARATTERI
50 REM
60 POKE65299,(PEEK(65299)AND3)+56
70 POKE65298,(PEEK(65298)AND251
80 REM
90 REM MODIFICA LA "@"
100 REM
110 FORK=0TO7 : READA : POKE14336+K,A : POKE14592+K,0 :
    NEXTK
120 REM
130 REM PULISCE LO SCHERMO E STAMPA 13 OMINI
140 REM
150 SCNCLR : PRINT"@@@@@@@@@@@@@@@@@@"
160 REM
170 REM ATTENDE LA PRESSIONE DI UN TASTO
180 REM
190 GETKEYA$
200 REM
210 REM RIPRISTINA PUNTATORI
220 REM
230 POKE65299,(PEEK(65299)AND3)+208
240 POKE65298,PEEK(65298)OR4
250 REM
260 REM DATI RELATIVI AL CARATTERE OMINO
270 REM
280 DATA 24,24,60,90,153,36,36,102
```

# PROGRAMMAZIONE

## Indicativo presente verbi in ARE

Non è facile, ma certo divertente, insegnare ad un computer la coniugazione di un verbo.

Proviamoci con questo programma che ha lo scopo di mettere in grado il tuo C16 di elaborare e visualizzare l'indicativo presente dei verbi regolari transitivi della 1<sup>a</sup> coniugazione (quelli dell'infinito in ARE).

```
10 DATA IO,O , TU, I, EGLI, A
20 DATA NOI, IAMO, VOI, ATE, ESSI, ANO
30 SCNCLR
40 INPUT "S INFINITO"; V$
50 R$=RIGHT$(V$,3)
60 IFR$ <>"ARE" THEN GOTO 40
70 N=LEN(V$) : C$=MID$(V$,N-3,1)
80 PRINT "C"
90 FORA=1 TO 6
100 READ A$,B$
110 R$=LEFT$(V$,N-3)
120 T$=LEFT$(B$,1)
130 IFCSS=T$ THEN R$=LEFT$(V$,N-4)
140 CHAR, 0, A, A$ : CHAR, 5, A, +R$+B$
150 NEXTA : PRINT : PRINT
160 INPUT "ANCORA (S/N) " ; R$
170 IFR$="S" THEN RUN
180 END
```

# PROGRAMMAZIONE



**L'errore è sempre in attesa: malignamente aspetta una tua distrazione, anche minima. Non dargli mai la possibilità di sogghignare. Impara la sintassi delle istruzioni. Realizza programmi in forma modulare, facilmente leggibili, e molto documentati. Nella digitazione, poi, metti la massima attenzione: anche una virgola al posto sbagliato, compromette la buona riuscita del programma.**

# VIDEOESERCIZI

Quali parole (costante stringa) verranno stampate?

```
10 READ A
20 READ B
30 FOR I = 1 TO A * B
40 READ A$
50 NEXT I
60 READ A$
70 PRINT A$
80 END
90 DATA 3, 1
100 DATA "VIDEO"
110 DATA "BASIC"
120 DATA "JACKSON"
```

```
130 DATA "SOFTIDEA"
140 DATA "COMMODORE"
150 DATA "PLUS 4"
160 DATA "C 16"
```

```
170 DATA "VIC 20"
180 DATA "6502"
190 DATA "6510"
200 DATA "VELOCE"
```

Molti discorsi di uomini politici sembrano avere la struttura impostata da questo programma cambiano opportunamente le linee data con vocabili come "congiuntura", "piattaforma" ecc.. ne otterrai degli esempi.

```
10 RESTORE
20 FOR I = 1 TO 2
30 LET A = INT (RND (1) * 10)
40 FOR K = 1 TO A
50 READ Z$ : NEXT K
60 IF I = 1 THEN READ A$
70 IF I = 2 THEN READ B$
80 RESTORE : FOR K = 1 TO 10 : READ Z$ : NEXT K : NEXT I
90 SCNCLR : PRINT A$; " "; B$
100 PRINT "S R R R R ANCOR (S/N)?" : REM S=HOME; R=I
110 GET Z$ : IF Z$ = " " THEN 110
120 IF Z$ = "S" THEN RUN
130 END
140 DATA "LA GIRAFFA", "L'UOMO", "LA MUCCA", "L'ASINO", "IL
CAVALLO", "L'ELEFANTE",
150 DATA "IL CANE", "IL GATTO", "LA GALLINA", "IL CONIGLIO"
160 DATA "SGHIGNAZZA", "SCOPPIA", "SALTICCHIA", "SVOLAZZA",
"RIMBALZA", FUGGE",
170 DATA "NITRISCE", "BARRISCE", "MUGGISCE", "SORRIDE"
```





**GRUPPO  
EDITORIALE  
JACKSON**